Pensieve header: Mathematica notebook for A Perturbed Alexander Invariant, with revised Rs.

In[ ]:=
```
SetDirectory["C:\\drorbn\\AcademicPensieve\\Projects\\APAI"];
```

The Program

tex

Two of the main reasons we like \$\rho_1\$ is that it is very easy to implement and even an unsophisticated implementation runs very fast. To highlight these points we include a full implementation here, a step-by-step run-through, and a demo run. We write in Mathematica~\cite{Wolfram:Mathematica}, and you can find the notebook displayed here at~\cite[APAI.nb]{Self}.

We start by loading the library \verb\$KnotTheory`\$~\cite{Bar-NatanMorrison:KnotTheory} (it is used here only for the list of knots that it contains, and to compute other invariants for comparisons). We also load minor conversion routine~\cite[Rot.nb / Rot.m]{Self} whose internal workings are irrelevant here.

pdf

In[ ]:=
```
Once[<< KnotTheory`; << Rot.m];
```

pdf

```
Loading KnotTheory` version of February 2, 2020, 10:53:45.2097.
Read more at http://katlas.org/wiki/KnotTheory.
```

pdf

```
Loading Rot.m from http://drorbn.net/APAI to compute rotation numbers.
```

tex

\needspace{50mm}
\subsection{The Program} This done, here is the full \$\rho_1\$ program:

pdf

In[ ]:=
```
R1[s_, i_, j_] := s (g_{ji} (g_{j+,j} + g_{j,j+} - g_{ij}) - g_{ii} (g_{j,j+} - 1) - 1/2 + θ (g_{i,i} g_{j,j} - g_{i,j} g_{j,i}));
ρ[K_] := Module[{Cs, φ, n, A, s, i, j, k, Δ, G, ρ1},
   {Cs, φ} = Rot[K]; n = Length[Cs];
   A = IdentityMatrix[2 n + 1];
   Cases[Cs, {s_, i_, j_} ⧴ (A⟦{i, j}, {i + 1, j + 1}⟧ += ( -T^s  T^s - 1 ; 0  -1 ))];
   Δ = T^((-Total[φ] - Total[Cs⟦All, 1⟧])/2) Det[A];
   G = Inverse[A];
   ρ1 = Sum_{k=1}^{n} R1 @@ Cs⟦k⟧ - Sum_{k=1}^{2 n} φ⟦k⟧ (g_{kk} - 1/2);
   Factor@{Δ, Δ^2 ρ1 /. α_^+ ⧴ α + 1 /. g_{α_,β_} ⧴ G⟦α, β⟧}];
```

tex

The program uses mostly the same symbols as the text, so even without any knowledge of Mathematica, the reader should be able to recognize at least formulas~\eqref{eq:A}, \eqref{eq:Delta}, and~\eqref{eq:rho1} within it. As a further hint we add that the variables \verb\$Cs\$ ends up storing the list of crossings in a knot \$K\$, where each crossing is stored as a triple \$(s,i,j)\$, where \$s\$, \$i\$, and \$j\$ have the same meaning as in~\eqref{eq:A}. The conversion routine \verb$Rot$ automatically produces \verb$Cs$, as well as a list φ of rotation numbers, given any other knot presentation known to the package \verb$KnotTheory`$.

tex

Note that the program outputs the ordered pair (Δ,ρ_1). The Alexander polynomial Δ is anyway computed internally, and we consider the aggregate (Δ,ρ_1) as more interesting than any of its pieces by itself.

Step-by-step Run-Through

tex

\subsection{A Step-by-Step Run-Through} We start by setting K to be the knot diagram on page~1 using the \verb$PD$ notation of \verb$KnotTheory`$~\cite{Bar-NatanMorrison:KnotTheory}. We then print \verb$Rot[K]$, which is a list of crossings followed by a list of rotation numbers:

pdf

In[ ]:=
```
K = PD[X[4, 2, 5, 1], X[2, 6, 3, 5], X[6, 4, 7, 3]];
Rot[K]
```

Out[ ]=
pdf

```
{{{1, 1, 4}, {1, 5, 2}, {1, 3, 6}}, {0, 0, 0, -1, 0, 0}}
```

tex

Next we set \verb$Cs$ and φ to be the list of crossings, and the list of rotation numbers, respectively.
\needspace{20mm}

pdf

In[ ]:=
```
{Cs, φ} = Rot[K]
```

Out[ ]=
pdf

```
{{{1, 1, 4}, {1, 5, 2}, {1, 3, 6}}, {0, 0, 0, -1, 0, 0}}
```

tex

We set \verb$n$ to be the number of crossings, \verb$A$ to be the $(2n+1)$-dimensional identity matrix, and then we iterate over \verb$c$ in \verb$Cs$, adding a block as in~\eqref{eq:A} for each crossing.

pdf

In[ ]:=
```
n = Length[Cs];
A = IdentityMatrix[2 n + 1];
Cases[Cs, {s_, i_, j_} :→ (A⟦{i, j}, {i + 1, j + 1}⟧ += ({{-T^s, T^s - 1}, {0, -1}}))];
```

tex

\needspace{30mm}
Here's what \verb$A$ comes out to be:

pdf

In[ ]:= **A // MatrixForm**

Out[ ]//MatrixForm=
pdf

$$\begin{pmatrix} 1 & -T & 0 & 0 & -1+T & 0 & 0 \\ 0 & 1 & -1 & 0 & 0 & 0 & 0 \\ 0 & 0 & 1 & -T & 0 & 0 & -1+T \\ 0 & 0 & 0 & 1 & -1 & 0 & 0 \\ 0 & 0 & -1+T & 0 & 1 & -T & 0 \\ 0 & 0 & 0 & 0 & 0 & 1 & -1 \\ 0 & 0 & 0 & 0 & 0 & 0 & 1 \end{pmatrix}$$

Here’s the same, in TeXForm:

In[ ]:= **A // MatrixForm // TeXForm**

Out[ ]//TeXForm=

```
\left(
\begin{array}{ccccccc}
 1 & -T & 0 & 0 & T-1 & 0 & 0 \\
 0 & 1 & -1 & 0 & 0 & 0 & 0 \\
 0 & 0 & 1 & -T & 0 & 0 & T-1 \\
 0 & 0 & 0 & 1 & -1 & 0 & 0 \\
 0 & 0 & T-1 & 0 & 1 & -T & 0 \\
 0 & 0 & 0 & 0 & 0 & 1 & -1 \\
 0 & 0 & 0 & 0 & 0 & 0 & 1 \\
\end{array}
\right)
```

tex

We set Δ to be the determinant of \verb$A$, with a correction as in~\eqref{eq:Delta}. So Δ is the Alexander polynomial of K.

In[ ]:= **Det[A]**

Out[ ]=

$1 - T + T^2$

pdf

In[ ]:= **Δ = T^((-Total[φ]-Total[Cs[[All,1]]])/2) Det[A]**

Out[ ]=
pdf

$$\frac{1 - T + T^2}{T}$$

tex

\needspace{30mm}
\verb$G$ is now the \verb$Inverse$ of \verb$A$:

pdf

In[ ]:=
```
G = Inverse[A];
G // MatrixForm
```

Out[ ]//MatrixForm=
pdf

$$\begin{pmatrix} 1 & \frac{T-T^2+T^3}{1-T+T^2} & 1 & \frac{T-T^2+T^3}{1-T+T^2} & 1 & \frac{T-T^2+T^3}{1-T+T^2} & 1 \\ 0 & 1 & \frac{1}{1-T+T^2} & \frac{T}{1-T+T^2} & \frac{T}{1-T+T^2} & \frac{T^2}{1-T+T^2} & 1 \\ 0 & 0 & \frac{1}{1-T+T^2} & \frac{T}{1-T+T^2} & \frac{T}{1-T+T^2} & \frac{T^2}{1-T+T^2} & 1 \\ 0 & 0 & \frac{1-T}{1-T+T^2} & \frac{1}{1-T+T^2} & \frac{1}{1-T+T^2} & \frac{T}{1-T+T^2} & 1 \\ 0 & 0 & \frac{1-T}{1-T+T^2} & \frac{T-T^2}{1-T+T^2} & \frac{1}{1-T+T^2} & \frac{T}{1-T+T^2} & 1 \\ 0 & 0 & 0 & 0 & 0 & 1 & 1 \\ 0 & 0 & 0 & 0 & 0 & 0 & 1 \end{pmatrix}$$

Here’s the same, in TeXForm:

In[ ]:=
```
G // MatrixForm // TeXForm
```

Out[ ]//TeXForm=
```
\left(
\begin{array}{ccccccc}
 1 & \frac{T^3-T^2+T}{T^2-T+1} & 1 & \frac{T^3-T^2+T}{T^2-T+1} & 1 & \frac{T^3-T^2+T}{T^2
 0 & 1 & \frac{1}{T^2-T+1} & \frac{T}{T^2-T+1} & \frac{T}{T^2-T+1} & \frac{T^2}{T^2-T+1}
 0 & 0 & \frac{1}{T^2-T+1} & \frac{T}{T^2-T+1} & \frac{T}{T^2-T+1} & \frac{T^2}{T^2-T+1}
 0 & 0 & \frac{1-T}{T^2-T+1} & \frac{1}{T^2-T+1} & \frac{1}{T^2-T+1} & \frac{T}{T^2-T+1}
 0 & 0 & \frac{1-T}{T^2-T+1} & \frac{T-T^2}{T^2-T+1} & \frac{1}{T^2-T+1} & \frac{T}{T^2-T
 0 & 0 & 0 & 0 & 0 & 1 & 1 \\
 0 & 0 & 0 & 0 & 0 & 0 & 1 \\
\end{array}
\right)
```

tex

\needspace{30mm}

It remains to blindly follow the two parts of Equation~\eqref{eq:rho1}:

pdf

In[ ]:=
$$\rho 1 = \sum_{k=1}^{n} R_1 \text{@@} Cs[[k]] - \sum_{k=1}^{2n} \varphi[[k]] (g_{kk} - 1/2)$$

Out[ ]=
pdf

$$-2 + g_{4,4} + \Theta\,(-g_{1,4}\, g_{4,1} + g_{1,1}\, g_{4,4}) - g_{1,1}\,(-1 + g_{4,4^+}) - (-1 + g_{2,2^+})\, g_{5,5} + \Theta\,(-g_{2,5}\, g_{5,2} + g_{2,2}\, g_{5,5}) + \Theta\,(-g_{3,6}\, g_{6,3} + g_{3,3}\, g_{6,6}) - g_{3,3}\,(-1 + g_{6,6^+}) + g_{2,5}\,(g_{2,2^+} - g_{5,2} + g_{2^+,2}) + g_{4,1}\,(-g_{1,4} + g_{4,4^+} + g_{4^+,4}) + g_{6,3}\,(-g_{3,6} + g_{6,6^+} + g_{6^+,6})$$

tex

We replace each ${\mathtt g}_{\alpha\beta}$ with the appropriate entry of \verb$G$:

pdf

In[ ]:=
$$\Delta^2 \rho 1 \text{ /. } \alpha\_^+ :\rightarrow \alpha + 1 \text{ /. } g_{\alpha\_,\beta\_} :\rightarrow G[[\alpha, \beta]]$$

Out[ ]=
pdf

$$\frac{\left(1 - T + T^2\right)^2 \left(-1 + \frac{T}{\left(1-T+T^2\right)^2} - \frac{-1+\frac{1}{1-T+T^2}}{1-T+T^2} + \frac{3\,\Theta}{1-T+T^2}\right)}{T^2}$$

tex

Finally, we output both \$\Delta\$ and \$\rho_1\$. We factor them just to put them in a nicer form:

pdf

In[ ]:=
```
Factor@{Δ, Δ^2 ρ1 /. α_^+ :→ α + 1 /. g_{α_,β_} :→ G⟦α, β⟧}
```

Out[ ]=
pdf

$$\left\{\frac{1 - T + T^2}{T}, -\frac{1 - 2\,T + 2\,T^2 - 2\,T^3 + T^4 - 3\,\theta + 3\,T\,\theta - 3\,T^2\,\theta}{T^2}\right\}$$

A Demo Run

tex

\subsection{A Demo Run} \label{ssec:Demo} Here are \$\Delta\$ and \$\rho_1\$ of all the knots with up to 6 crossings (a table up to 10 crossings is printed at~\cite{PG}:

pdf

In[ ]:=
```
TableForm[Table[Join[{K}, {Factor@Coefficient[ρ[K]⟦2⟧, θ]}, ρ[K]],
  {K, AllKnots[{3, 6}]}], TableAlignments → Center]
```

pdf

KnotTheory: Loading precomputed data in PD4Knots`.

Out[ ]//TableForm=
pdf

Knot[3, 1]	$-3\,T\,\left(1 - T + T^2\right)$	$\frac{1-T+T^2}{T}$	$-\frac{-1+2\,T}{}$
Knot[4, 1]	$\frac{2\,(-1+T)^2\,\left(1-3\,T+T^2\right)}{T^2}$	$-\frac{1-3\,T+T^2}{T}$	
Knot[5, 1]	$-5\,T\,\left(1 - T + T^2 - T^3 + T^4\right)$	$\frac{1-T+T^2-T^3+T^4}{T^2}$	$-\frac{-2+4\,T-5\,T^2+6\,T^3-6\,T^4+6}{}$
Knot[5, 2]	$-\frac{\left(2-3\,T+2\,T^2\right)\,\left(3-8\,T+10\,T^2\right)}{T}$	$\frac{2-3\,T+2\,T^2}{T}$	$-\frac{-5+14\,T-18\,T^2+14}{}$
Knot[6, 1]	$-\frac{2\,(-2+T)\,(-1+2\,T)\,\left(1-6\,T+4\,T^2\right)}{T}$	$-\frac{(-2+T)\,(-1+2\,T)}{T}$	$-\frac{-1+6\,T-10\,T^2+6}{}$
Knot[6, 2]	$-\frac{\left(1-3\,T+3\,T^2-3\,T^3+T^4\right)\,\left(-1+T-T^2-5\,T^3+4\,T^4\right)}{T^3}$	$-\frac{1-3\,T+3\,T^2-3\,T^3+T^4}{T^2}$	$-\frac{-1+6\,T-13\,T^2+16\,T^3-16\,T^4+16\,T^5-13\,T^6+6}{}$
Knot[6, 3]	$-\frac{(-1+T)\,(1+T)\,\left(1-T+T^2\right)\,\left(1-3\,T+5\,T^2-3\,T^3+T^4\right)}{T^4}$	$\frac{1-3\,T+5\,T^2-3\,T^3+T^4}{T^2}$	$-\frac{(-1+T)\,(}{}$

tex

Some comments are in order:
\begin{itemize}
\item \$\rho_1\$ flips its sign when switching to the mirror of a knot. Indeed in~\eqref{eq:rho1} both \$R_1(c)\$ and \$\varphi_k\$ flip sign under reflection in a plane perpendicular to the plane of the knot diagram. Hence \$\rho_1\$ vanishes on amphicheiral knots, such as \$4_1\$ and \$6_3\$ above.
\item \$\rho_1\$ seems to always be divisible by \$(T-1)^2\$ and seems to always be palindromic (\$\rho_1(T)=\rho_1(T^{-1})\$). We are not sure why this is so.
\end{itemize}

tex

\begin{figure}
\[\resizebox{6in}{!}{\input{figs/GST48-Marked.pdf_t}} \]
\caption{A 48-crossing knot from~\cite{GompfScharlemannThompson:Counterexample}.} \label{-
fig:GST48}

\end{figure}

Next is one of our favourites, a knot from~\cite{GompfScharlemannThompson:Counterexample} (see Figure~\ref{fig:GST48}), which is a potential counterexample to the ribbon$=$slice conjecture~\cite{-Fox:Problems}. It takes about two minutes to compute ρ_1 for this 48 crossing knot (note that Mathematica prints \verb$Timing$ information is seconds, and that this information is highly dependent on the CPU used, how loaded it is, and even on its temperature at the time of the computation):

pdf

In[ ]:=

$$\text{Timing}@\rho\left[\text{EPD}\left[X_{14,1}, \overline{X}_{2,29}, X_{3,40}, X_{43,4}, \overline{X}_{26,5}, X_{6,95}, X_{96,7}, X_{13,8}, \overline{X}_{9,28}, X_{10,41}, X_{42,11}, \overline{X}_{27,12}, X_{30,15}, \overline{X}_{16,61}, \overline{X}_{17,72}, \overline{X}_{18,83}, X_{19,34}, \overline{X}_{89,20}, \overline{X}_{21,92}, \overline{X}_{79,22}, \overline{X}_{68,23}, \overline{X}_{57,24}, \overline{X}_{25,56}, X_{62,31}, X_{73,32}, X_{84,33}, \overline{X}_{50,35}, X_{36,81}, X_{37,70}, X_{38,59}, \overline{X}_{39,54}, X_{44,55}, X_{58,45}, X_{69,46}, X_{80,47}, X_{48,91}, X_{90,49}, X_{51,82}, X_{52,71}, X_{53,60}, \overline{X}_{63,74}, \overline{X}_{64,85}, \overline{X}_{76,65}, \overline{X}_{87,66}, \overline{X}_{67,94}, \overline{X}_{75,86}, \overline{X}_{88,77}, \overline{X}_{78,93}\right]\right]$$

Out[ ]=

pdf

$$\Big\{138.266, \Big\{-\frac{\left(-1+2T-T^2-T^3+2T^4-T^5+T^8\right)\left(-1+T^3-2T^4+T^5+T^6-2T^7+T^8\right)}{T^8},$$
$$\frac{1}{T^{20}}\big(5T^4-28T^5+74T^6-116T^7+99T^8+6T^9-144T^{10}+158T^{11}+120T^{12}-582T^{13}+758T^{14}-326T^{15}-382T^{16}+732T^{17}-533T^{18}+134T^{19}+50T^{20}+134T^{21}-533T^{22}+732T^{23}-382T^{24}-326T^{25}+758T^{26}-582T^{27}+120T^{28}+158T^{29}-144T^{30}+6T^{31}+99T^{32}-116T^{33}+74T^{34}-28T^{35}+5T^{36}+3\theta-8T\theta+T^2\theta+23T^3\theta-50T^4\theta+98T^5\theta-221T^6\theta+359T^7\theta-301T^8\theta+16T^9\theta+111T^{10}\theta+42T^{11}\theta+363T^{12}\theta-1862T^{13}\theta+2835T^{14}\theta-1008T^{15}\theta-2660T^{16}\theta+4185T^{17}\theta-1862T^{18}\theta-1149T^{19}\theta+1500T^{20}\theta-193T^{21}\theta-53T^{22}\theta-681T^{23}\theta+633T^{24}\theta+106T^{25}\theta-153T^{26}\theta-462T^{27}\theta+553T^{28}\theta+39T^{29}\theta-418T^{30}\theta+256T^{31}\theta-41T^{32}\theta+59T^{33}\theta-119T^{34}\theta+82T^{35}\theta-22T^{36}\theta+T^{37}\theta\big)\Big\}\Big\}$$

tex

\subsection{The Separation Power of ρ_1} Let us check how powerful ρ_1 is on knots with up to 12 crossings:

pdf

```
{NumberOfKnots[{3, 12}],
 Length@Union@Table[ρ[K], {K, AllKnots[{3, 12}]}],
 Length@Union@Table[{HOMFLYPT[K], Kh[K]}, {K, AllKnots[{3, 12}]}]}
```

Out[ ]=

pdf

```
{2977, 2882, 2785}
```

tex

So the pair (Δ,ρ_1) attains 2,882 distinct values on the 2,977 prime knots with up to 12 crossings (a deficit of 95), whereas the pair $(H,Kh)=$ (HOMFLYPT polynomial, Khovanov Homology) attains only 2,785 distinct values on the same knots (a deficit of 192).

tex

In our spare time we computed all of these invariants on all the prime knots with up to 14 crossings. On these 59,937 knots the pair (Δ,ρ_1) attains 53,684 distinct values (a deficit of 6,253) whereas the pair (H,Kh) attains only 49,149 distinct values on the same knots (a deficit of 10,788).

tex

Hence the pair (Δ,ρ_1), computable in polynomial time by simple programs, seems stronger than the pair (H,Kh), which is more difficult to program and (for all we know) cannot be computed in polynomial time. We are not aware of another poly-time invariant as strong as the pair (Δ,ρ_1).

The g-Rules

exec

```
nb2tex$TeXFileName = "gRules.tex";
```

pdf

In[ ]:=

$$\delta_{i\_,j\_} := \mathrm{If}[i === j, 1, 0];$$
$$\mathrm{gRules}_{s\_,i\_,j\_} := \{g_{i\beta\_} \mapsto \delta_{i\beta} + T^s g_{i^+,\beta} + (1 - T^s)\, g_{j^+,\beta},\ g_{j\beta\_} \mapsto \delta_{j\beta} + g_{j^+,\beta},$$
$$g_{\alpha\_,i} \mapsto T^{-s}\,(g_{\alpha,i^+} - \delta_{\alpha,i^+}),\ g_{\alpha\_j} \mapsto g_{\alpha,j^+} - (1 - T^s)\, g_{\alpha i} - \delta_{\alpha,j^+}\}$$
$$(\alpha\_^+)^+ := \alpha^{\text{"++"}};\quad (\text{* this is for cosmetic reasons only *})$$

Invariance Under R3

exec

```
nb2tex$TeXFileName = "Invariance-R3-Short.tex";
```

pdf

In[ ]:=

$$\mathrm{lhs} = R_1[1, j, k] + R_1[1, i, k^+] + R_1[1, i^+, j^+]\ //.\ \mathrm{gRules}_{1,j,k} \cup \mathrm{gRules}_{1,i,k^+} \cup \mathrm{gRules}_{1,i^+,j^+};$$
$$\mathrm{rhs} = R_1[1, i, j] + R_1[1, i^+, k] + R_1[1, j^+, k^+]\ //.\ \mathrm{gRules}_{1,i,j} \cup \mathrm{gRules}_{1,i^+,k} \cup \mathrm{gRules}_{1,j^+,k^+};$$
$$\mathrm{Simplify}[\mathrm{lhs} == \mathrm{rhs}]$$

Out[ ]=
pdf

$$\frac{(-1 + T^2)\ \theta\ (g_{j^{++},j^{++}}\, g_{k^{++},i^{++}} - g_{j^{++},i^{++}}\, g_{k^{++},j^{++}})}{T} == 0$$

exec

```
nb2tex$TeXFileName = "Invariance-R3.tex";
```

tex

\needspace{30mm}

pdf

In[ ]:=

$$\texttt{lhs} = \texttt{Simplify}\left[R_1[1, j, k] + R_1[1, i, k^+] + R_1[1, i^+, j^+] \;//.\; \texttt{gRules}_{1,j,k} \cup \texttt{gRules}_{1,i,k^+} \cup \texttt{gRules}_{1,i^+,j^+}\right]$$

Out[ ]=
pdf

$$\begin{aligned}
&-\frac{3}{2} + \frac{(-1+T)\, g^2_{j^{++},i^{++}}}{T} + g_{j^{++},j^{++}} - g_{k^{++},i^{++}} - g_{i^{++},k^{++}}\, g_{k^{++},i^{++}} - \theta\, g_{i^{++},k^{++}}\, g_{k^{++},i^{++}} - g_{j^{++},j^{++}}\, g_{k^{++},i^{++}} + \\
&\frac{g_{j^{++},j^{++}}\, g_{k^{++},i^{++}}}{T} - \frac{\theta\, g_{j^{++},j^{++}}\, g_{k^{++},i^{++}}}{T^2} + \frac{\theta\, g_{j^{++},j^{++}}\, g_{k^{++},i^{++}}}{T} + g^2_{k^{++},i^{++}} - \frac{g^2_{k^{++},i^{++}}}{T} - g_{k^{++},j^{++}} - \\
&g_{j^{++},j^{++}}\, g_{k^{++},j^{++}} + \frac{g_{j^{++},j^{++}}\, g_{k^{++},j^{++}}}{T} - g_{j^{++},k^{++}}\, g_{k^{++},j^{++}} - \theta\, g_{j^{++},k^{++}}\, g_{k^{++},j^{++}} + 2\, g_{k^{++},i^{++}}\, g_{k^{++},j^{++}} - \\
&\frac{2\, g_{k^{++},i^{++}}\, g_{k^{++},j^{++}}}{T} + g^2_{k^{++},j^{++}} - \frac{g^2_{k^{++},j^{++}}}{T} - \frac{1}{T^2} g_{j^{++},i^{++}} \Big(T^2 + T^2\, (1+\theta)\, g_{i^{++},j^{++}} - 2\, T^2\, g_{j^{++},j^{++}} + \\
&\quad g_{k^{++},i^{++}} - 2\, T\, g_{k^{++},i^{++}} + T^2\, g_{k^{++},i^{++}} - T\, g_{k^{++},j^{++}} + T^2\, g_{k^{++},j^{++}} - \theta\, g_{k^{++},j^{++}} + T\, \theta\, g_{k^{++},j^{++}}\Big) - \\
&g_{j^{++},j^{++}}\, g_{k^{++},k^{++}} + \theta\, g_{j^{++},j^{++}}\, g_{k^{++},k^{++}} + 2\, g_{k^{++},i^{++}}\, g_{k^{++},k^{++}} + 2\, g_{k^{++},j^{++}}\, g_{k^{++},k^{++}} + \\
&g_{i^{++},i^{++}} \left(2 + \left(-1 + \frac{1}{T}\right) g_{j^{++},i^{++}} + (-1+\theta)\, g_{j^{++},j^{++}} + \frac{g_{k^{++},i^{++}}}{T^2} - \frac{g_{k^{++},i^{++}}}{T} - g_{k^{++},k^{++}} + \theta\, g_{k^{++},k^{++}}\right)
\end{aligned}$$

pdf

In[ ]:=

$$\texttt{rhs} = \texttt{Simplify}\left[R_1[1, i, j] + R_1[1, i^+, k] + R_1[1, j^+, k^+] \;//.\; \texttt{gRules}_{1,i,j} \cup \texttt{gRules}_{1,i^+,k} \cup \texttt{gRules}_{1,j^+,k^+}\right]$$

Out[ ]=
pdf

$$\begin{aligned}
&-\frac{1}{2\, T^2} \\
&\Big(-2\, (-1+T)\, T\, g^2_{j^{++},i^{++}} + 2\, g_{j^{++},i^{++}} \Big(T^2 + T^2\, (1+\theta)\, g_{i^{++},j^{++}} - 2\, T^2\, g_{j^{++},j^{++}} + g_{k^{++},i^{++}} - 2\, T\, g_{k^{++},i^{++}} + T^2\, g_{k^{++},i^{++}} - \\
&\quad T\, g_{k^{++},j^{++}} + T^2\, g_{k^{++},j^{++}} + T\, \theta\, g_{k^{++},j^{++}} - T^2\, \theta\, g_{k^{++},j^{++}}\Big) + 2\, g_{i^{++},i^{++}} \\
&\quad \Big(-2\, T^2 + (-1+T)\, T\, g_{j^{++},i^{++}} - T^2\, (-1+\theta)\, g_{j^{++},j^{++}} - g_{k^{++},i^{++}} + T\, g_{k^{++},i^{++}} + T^2\, g_{k^{++},k^{++}} - T^2\, \theta\, g_{k^{++},k^{++}}\Big) + \\
&\quad T \Big(3\, T - 2\, (-1+T)\, g^2_{k^{++},i^{++}} + 2\, T\, g_{k^{++},j^{++}} + 2\, T\, g_{j^{++},k^{++}}\, g_{k^{++},j^{++}} + 2\, T\, \theta\, g_{j^{++},k^{++}}\, g_{k^{++},j^{++}} + 2\, g^2_{k^{++},j^{++}} - \\
&\quad 2\, T\, g^2_{k^{++},j^{++}} - 4\, T\, g_{k^{++},j^{++}}\, g_{k^{++},k^{++}} + 2\, g_{k^{++},i^{++}} \Big(T + T\, (1+\theta)\, g_{i^{++},k^{++}} - 2\, (-1+T)\, g_{k^{++},j^{++}} - 2\, T\, g_{k^{++},k^{++}}\Big) + \\
&\quad 2\, g_{j^{++},j^{++}} \Big((-1+T)\, (1+\theta)\, g_{k^{++},i^{++}} + (-1+T)\, g_{k^{++},j^{++}} + T\, (-1 - (-1+\theta)\, g_{k^{++},k^{++}})\Big)\Big)\Big)
\end{aligned}$$

pdf

In[ ]:=

```
lhs == rhs
```

Out[ ]=
pdf

$$-\frac{3}{2}+\frac{(-1+T)\,g^2_{j^{++},i^{++}}}{T}+g_{j^{++},j^{++}}-g_{k^{++},i^{++}}-g_{i^{++},k^{++}}\,g_{k^{++},i^{++}}-\Theta\,g_{i^{++},k^{++}}\,g_{k^{++},i^{++}}-g_{j^{++},j^{++}}\,g_{k^{++},i^{++}}+$$

$$\frac{g_{j^{++},j^{++}}\,g_{k^{++},i^{++}}}{T}-\frac{\Theta\,g_{j^{++},j^{++}}\,g_{k^{++},i^{++}}}{T^2}+\frac{\Theta\,g_{j^{++},j^{++}}\,g_{k^{++},i^{++}}}{T}+g^2_{k^{++},i^{++}}-\frac{g^2_{k^{++},i^{++}}}{T}-g_{k^{++},j^{++}}-$$

$$g_{j^{++},j^{++}}\,g_{k^{++},j^{++}}+\frac{g_{j^{++},j^{++}}\,g_{k^{++},j^{++}}}{T}-g_{j^{++},k^{++}}\,g_{k^{++},j^{++}}-\Theta\,g_{j^{++},k^{++}}\,g_{k^{++},j^{++}}+2\,g_{k^{++},i^{++}}\,g_{k^{++},j^{++}}-$$

$$\frac{2\,g_{k^{++},i^{++}}\,g_{k^{++},j^{++}}}{T}+g^2_{k^{++},j^{++}}-\frac{g^2_{k^{++},j^{++}}}{T}-\frac{1}{T^2}\,g_{j^{++},i^{++}}\left(T^2+T^2\,(1+\Theta)\,g_{i^{++},j^{++}}-2\,T^2\,g_{j^{++},j^{++}}+\right.$$

$$\left.g_{k^{++},i^{++}}-2\,T\,g_{k^{++},i^{++}}+T^2\,g_{k^{++},i^{++}}-T\,g_{k^{++},j^{++}}+T^2\,g_{k^{++},j^{++}}-\Theta\,g_{k^{++},j^{++}}+T\,\Theta\,g_{k^{++},j^{++}}\right)-$$

$$g_{j^{++},j^{++}}\,g_{k^{++},k^{++}}+\Theta\,g_{j^{++},j^{++}}\,g_{k^{++},k^{++}}+2\,g_{k^{++},i^{++}}\,g_{k^{++},k^{++}}+2\,g_{k^{++},j^{++}}\,g_{k^{++},k^{++}}+$$

$$g_{i^{++},i^{++}}\left(2+\left(-1+\frac{1}{T}\right)g_{j^{++},i^{++}}+(-1+\Theta)\,g_{j^{++},j^{++}}+\frac{g_{k^{++},i^{++}}}{T^2}-\frac{g_{k^{++},i^{++}}}{T}-g_{k^{++},k^{++}}+\Theta\,g_{k^{++},k^{++}}\right)==$$

$$-\frac{1}{2\,T^2}\left(-2\,(-1+T)\,T\,g^2_{j^{++},i^{++}}+2\,g_{j^{++},i^{++}}\left(T^2+T^2\,(1+\Theta)\,g_{i^{++},j^{++}}-2\,T^2\,g_{j^{++},j^{++}}+g_{k^{++},i^{++}}-\right.\right.$$

$$\left.2\,T\,g_{k^{++},i^{++}}+T^2\,g_{k^{++},i^{++}}-T\,g_{k^{++},j^{++}}+T^2\,g_{k^{++},j^{++}}+T\,\Theta\,g_{k^{++},j^{++}}-T^2\,\Theta\,g_{k^{++},j^{++}}\right)+2\,g_{i^{++},i^{++}}$$

$$\left(-2\,T^2+(-1+T)\,T\,g_{j^{++},i^{++}}-T^2\,(-1+\Theta)\,g_{j^{++},j^{++}}-g_{k^{++},i^{++}}+T\,g_{k^{++},i^{++}}+T^2\,g_{k^{++},k^{++}}-T^2\,\Theta\,g_{k^{++},k^{++}}\right)+$$

$$T\left(3\,T-2\,(-1+T)\,g^2_{k^{++},i^{++}}+2\,T\,g_{k^{++},j^{++}}+2\,T\,g_{j^{++},k^{++}}\,g_{k^{++},j^{++}}+2\,T\,\Theta\,g_{j^{++},k^{++}}\,g_{k^{++},j^{++}}+2\,g^2_{k^{++},j^{++}}-\right.$$

$$2\,T\,g^2_{k^{++},j^{++}}-4\,T\,g_{k^{++},j^{++}}\,g_{k^{++},k^{++}}+2\,g_{k^{++},i^{++}}\left(T+T\,(1+\Theta)\,g_{i^{++},k^{++}}-2\,(-1+T)\,g_{k^{++},j^{++}}-2\,T\,g_{k^{++},k^{++}}\right)+$$

$$\left.\left.2\,g_{j^{++},j^{++}}\left((-1+T)\,(1+\Theta)\,g_{k^{++},i^{++}}+(-1+T)\,g_{k^{++},j^{++}}+T\,(-1-(-1+\Theta)\,g_{k^{++},k^{++}})\right)\right)\right)$$

Invariance Under R2c

exec

```
nb2tex$TeXFileName = "Invariance-R2c.tex";
```

pdf

In[ ]:=

$$\text{Simplify}\left[R_1[-1,\,i,\,j^+]+R_1[1,\,i^+,\,j]-\left(g_{j^+,j^+}-1/2\right)\right]$$

$$\text{lhs}=\text{Simplify}\left[R_1[-1,\,i,\,j^+]+R_1[1,\,i^+,\,j]-\left(g_{j^+,j^+}-1/2\right)\ \text{//.}\ \text{gRules}_{-1,i,j^+}\cup\text{gRules}_{1,i^+,j}\right]$$

Out[ ]=
pdf

$$\frac{1}{2}-\left(-1+g_{j,j^+}\right)g_{i^+,i^+}+\Theta\left(-g_{j,i^+}\,g_{i^+,j}+g_{j,j}\,g_{i^+,i^+}\right)+g_{j,i^+}\left(g_{j,j^+}-g_{i^+,j}+g_{j^+,j}\right)+$$

$$g_{i,i}\left(-1+g_{j^+,j^{++}}\right)-g_{j^+,i}\left(-g_{i,j^+}+g_{j^{++},j^+}+g_{j^+,j^{++}}\right)-g_{j^+,j^+}+\Theta\left(g_{i,j^+}\,g_{j^+,i}-g_{i,i}\,g_{j^+,j^+}\right)$$

Out[ ]=
pdf

$$\frac{1}{2}-\frac{(-1+T)\,\Theta\,g_{j^{++},i^{++}}}{T}-g_{j^{++},j^{++}}$$

Invariance Under R1l

exec

```
nb2tex$TeXFileName = "Invariance-R1l.tex";
```

pdf

In[ ]:=

$$\text{lhs1} = R_1[1, i^+, i] - (g_{i^+,i^+} - 1/2)$$
$$\text{lhs2} = \text{lhs1} \;//.\; \{g_{i^+,\beta\_} \mapsto T^{-1}\,\delta_{i^+,\beta} + g_{i^{++},\beta},\; g_{i,\beta\_} \mapsto \delta_{i,\beta} + g_{i^+,\beta}\}$$
$$\text{Simplify}[\text{lhs2}]$$

Out[ ]=
pdf

$$g_{i,i^+}^2 - g_{i^+,i^+} - (-1 + g_{i,i^+})\, g_{i^+,i^+} + \ominus\,(-g_{i,i^+}\, g_{i^+,i} + g_{i,i}\, g_{i^+,i^+})$$

Out[ ]=
pdf

$$-\frac{1}{T} - g_{i^{++},i^+} - \left(-1 + \frac{1}{T} + g_{i^{++},i^+}\right)\left(\frac{1}{T} + g_{i^{++},i^+}\right) + \left(\frac{1}{T} + g_{i^{++},i^+}\right)^2 + \ominus\left(-g_{i^{++},i}\left(\frac{1}{T} + g_{i^{++},i^+}\right) + (1 + g_{i^{++},i})\left(\frac{1}{T} + g_{i^{++},i^+}\right)\right)$$

Out[ ]=
pdf

$$\ominus\left(\frac{1}{T} + g_{i^{++},i^+}\right)$$

R1r, R2b, and Sw$^+$.

exec

```
nb2tex$TeXFileName = "Invariance-Rest.tex";
```

pdf

In[ ]:=

$$\text{Simplify}\Big[R_1[1, i, i^+] + (g_{i^+,i^+} - 1/2) \;//.\; \{ \quad (* \text{ R1r } *)$$
$$g_{i\beta\_} \mapsto \delta_{i\beta} + T\, g_{i^+,\beta} + (1 - T)\, g_{i^{++},\beta},\; g_{i^+,\beta\_} \mapsto \delta_{i^+,\beta} + g_{i^{++},\beta},$$
$$g_{\alpha\_,i} \mapsto T^{-1}\,(g_{\alpha,i^+} - \delta_{\alpha,i^+}),\; g_{\alpha\_,i^+} \mapsto T\, g_{\alpha,i^{++}} - (1 - T)\,\delta_{\alpha,i^+} - T\,\delta_{\alpha,i^{++}} \}\Big]$$

Out[ ]=
pdf

$$\ominus\, g_{i^{++},i^{++}}$$

tex

\noindent(Note that the version of the g-rules we used above easily follows from~\eqref{eq:-CounterRules}).

pdf

In[ ]:=

$$\text{Simplify}\Big[R_1[1, i, j] + R_1[-1, i^+, j^+] \;//.\; \text{gRules}_{1,i,j} \cup \text{gRules}_{-1,i^+,j^+}\Big] \quad (* \text{ R2b } *)$$

Out[ ]=
pdf

0

pdf

In[ ]:=

$$(g_{i,i} - 1/2) + (g_{j,j} - 1/2) - (g_{i^+,i^+} - 1/2) - (g_{j^+,j^+} - 1/2) \;//.\; \text{gRules}_{1,i,j} \quad (* \text{ Sw}^+ *)$$

Out[ ]=
pdf

0